

1 BC BNF

program : ϵ | program rule '.'
rule : fact | law | query
fact : act_fact | flu_fact
law : static_law | dynamic_law | inertial_law
 | default_law | impossible_law | nonexecutable_law

act_fact : '<action>' action_list
flu_fact : '<fluent>' literal_list
 | '<defined fluent>' literal_list

static_law : formula '<if>' formula
 | formula '<ifcons>' formula
 | formula '<if>' formula '<ifcons>' formula

dynamic_law : formula '<after>' action_formula
 | formula '<after>' action_formula '<ifcons>' formula
 | action_list '<causes>' formula
 | action_list '<causes>' formula '<if>' formula

inertial_law : '<inertial>' literal_list
default_law : '<default>' formula
 | '<default>' formula '<if>' formula
 | '<default>' formula '<after>' action_formula
impossible_law : '<impossible>' formula
 | '<impossible>' formula '<ifcons>' formula
nonexecutable_law : '<nonexecutable>' action_list
 | '<nonexecutable>' action_list '<if>' formula
 | '<nonexecutable>' action_list '<ifcons>' formula
 | '<nonexecutable>' action_list '<if>' formula '<ifcons>' formula

formula : atom | formula ',' atom
action_formula : atom | action_atom | action_formula ',' atom
 | action_formula ',' action_atom
action_list : action | action_list ',' action
atom : literal | '-' literal | literal '=' IDENTIFIER
literal_list : literal | literal_list ',' literal
action_atom : action | '-' action
literal : IDENTIFIER
action : IDENTIFIER